


Kör rajzolása 1

Rutin készítése kör rajzolásához, a kör origójának és sugarának megadásával.



A robot programozási nyelvek mindegyike tartalmaz a kör rajzolásához használható utasításokat. Ezen utasítások működési elve nagyon hasonló egymáshoz. A lényese, hogy a kört – ami programnyelvtől függően félkör, vagy attól kisebb vagy nagyobb körív, de semmiképpen nem teljes kör – három pontos szerkesztési módszerrel hozza létre. A robot aktuális pozíciójából kiindulva, egy köztes ponton keresztül a végpontig tartó körívet szerkeszt, melyen a robot TCP végig halad. Teljes kört jellemzően két egymás után elhelyezett utasítással lehet létrehozni.

Az ABB rapid nyelvén ez a következőképpen néz ki:

MoveC

Az utasítás paraméter készlete a mozgásutasításoknál már megszokottakon túl, tartalmaz egy köztes pontra vonatkozó rotarget típusú paramétert is.

```
MoveC C2,C3,v1000,z1,tool0;
```

Az a módszer kiválóan alkalmas akkor, ha egy felület mentén szeretnénk haladni és annak egyes pontjait letapogatva tanítjuk be a robot számára szükséges pontokat.

Mi történik abban az esetben, ha nem egy körív megadásával szeretnénk a kört megrajzolni, hanem a kör alapvető geometriai paramétereinek segítségével, mondjuk az origó helyének és a kör sugarának megadásával. Nem tehetünk mást, mint készítsünk egy rövid programot, rutint, mely a szükséges számítások elvégzése után összeállítja a körív rajzolási utasításokat.

Mivel a RAPID egy teljes értékű programnyelv a megoldásra számos lehetőség kínálkozik. A legalapvetőbb, ha a kívánt paramétereket manuálisan módosítva hozzuk létre a megfelelő robtarget változókat, és azokból állítjuk össze a MoveC utasításokat. A kód kissé terjedelmes lesz, de minden lépés jól következtető és lehetőség van a robtarget változó működésének megismerésére is.

A rutint, a rajzolni kívánt kör sugarának és a kör középpontjának megadásával együtt hívjuk meg, (ez a paraméteres rutin hívás) ezzel biztosítva, hogy a rutin felhasználható legyen tetszőleges középpontú és sugarú kör rajzolására. Természetesen a paraméter lista bővíthető további elemekkel is, ahogy ez látható lesz a későbbiekben.

A rutin hívás a következőképpen néz ki.

```
VAR num direction:=0;
VAR num radius:=0;
VAR robtarget origo=[[0,0,0],[1,0,0,0],[0,0,0,0],[9E9,9E9,9E9,9E9,9E9,9E9]];

PROC main()
  circle origo,radius,direction;
ENDPROC

PROC circle(var robtarget o,var num r,var num d)
ENDPROC
```

A kör rajzoló rutinon belül a paraméterek manuális módosítása pedig,

```

PROC circle(var robtarget o,var num r,var num d)

  VAR robtarget C1:=[[0,0,0],[1,0,0],[0,0,0],[9E9,9E9,9E9,9E9,9E9,9E9]];
  VAR robtarget C2:=[[0,0,0],[1,0,0],[0,0,0],[9E9,9E9,9E9,9E9,9E9,9E9]];
  VAR robtarget C3:=[[0,0,0],[1,0,0],[0,0,0],[9E9,9E9,9E9,9E9,9E9,9E9]];
  VAR robtarget C4:=[[0,0,0],[1,0,0],[0,0,0],[9E9,9E9,9E9,9E9,9E9,9E9]];

  C1:=o;
  C2:=o;
  C3:=o;
  C4:=o;

  IF d=1 THEN
    C1.trans.x:=C1.trans.x+r;
    C2.trans.Y:=C2.trans.Y+r;
    C3.trans.x:=C3.trans.x-r;
    C4.trans.Y:=C4.trans.Y-r;
  ENDIF
  IF d=2 THEN
    C1.trans.z:=C1.trans.z+r;
    C2.trans.Y:=C2.trans.Y+r;
    C3.trans.z:=C3.trans.z-r;
    C4.trans.Y:=C4.trans.Y-r;
  ENDIF

  MoveJ C1,v1000,fine,tool0;
  MoveC C2,C3,v1000,z1,tool0;
  MoveC C4,C1,v1000,fine,tool0;

ENDPROC

```

A **MoveC** utasítások megfelelő felparaméterezése egy jóval rövidebb kód segítségével is megtehető, de itt a művelet egyes lépései nehezebben választhatók szét.

```

PROC circleB(var robtarget ori,var num r,var num d)

  IF d=1 THEN
    MoveJ Offs(ori,r,0,0),v1000,fine,tool0;
    MoveC Offs(ori,0,r,0),Offs(ori,-r,0,0),v1000,z1,tool0;
    MoveC Offs(ori,0,-r,0),Offs(ori,r,0,0),v1000,fine,tool0;
  ENDIF
  IF d=2 THEN
    MoveJ Offs(ori,0,0,r),v1000,fine,tool0;
    MoveC Offs(ori,0,r,0),Offs(ori,0,0,-r),v1000,z1,tool0;
    MoveC Offs(ori,0,-r,0),Offs(ori,0,0,r),v1000,fine,tool0;
  ENDIF

ENDPROC

```

A fenti programkód az Offs (offset) függvényt használja a robtarget paramétereinek módosítására. A két megoldás közti különbség a szemmel láthatókon kívül, hogy az offset függvénnyel módosított robtargrt csak az utasítás lefutásáig elérhető, utána már csak az eredeti paraméterekkel rendelkező ORI robtarget áll rendelkezésre.